

A GEODETIC APPROACH TO GRAVITY DATA REDUCTION FOR GEOPHYSICS

W. E. FEATHERSTONE¹ and M.C. DENTITH²

1. School of Surveying and Land Information, Curtin University of Technology, Perth, Australia
2. Department of Geology and Geophysics, The University of Western Australia, Perth, Australia

Abstract - The currently adopted approach to reduce observed gravity data for geophysical purposes includes several approximations. These were originally used to reduce computational effort, but have remained standard practice even though the required computing power is now readily available. In contrast, more precise gravity reductions are routinely employed in physical geodesy. The difference between simple Bouguer gravity anomalies derived using the geophysical and geodetic approaches can reach several tens of μms^{-2} . The geodetic reductions include a more accurate calculation of normal gravity as a function of latitude, and a free air correction that accounts for the non-sphericity of the figure of the Earth. Also important, especially given the advent of Global Positioning System coordination of gravity surveys, is the need to ensure that the correct vertical and horizontal coordinate systems are used for the gravity reduction procedure. Errors associated with the use of non-geocentric horizontal coordinates and ellipsoidal heights are significant when compared with the accuracy of an individual gravity measurement. A generalised gravity reduction program and a coordinate transformation program are presented which can be employed to reduce geophysical data in a geodetic manner.

Key Words: gravity, geodesy, coordinate transformations, data processing, geophysics.

INTRODUCTION

Spatial variations in the gravitational attraction of the Earth's mass are used in both geodesy and geophysics. Geodesy uses these variations to study departures of the figure of the Earth from a simplified ellipsoidal shape, whereas geophysics uses them to identify lateral density contrasts in the sub-surface and hence infer the local geology. As the requirements of these two disciplines are somewhat different, so are their approaches to convert gravity observations to a useable form.

Geodesy tends to use more precise reduction methods, whereas geophysics uses simplified versions of the same formulae. The adoption of these simplifications is mainly historical, as they were originally designed to reduce the computational effort. Nowadays, however, the increases in readily available computing power enable the more precise gravity reduction techniques used in geodesy to be routinely applied in geophysics. Existing geophysical reduction methods are often adequate for the localised surveys used in exploration for resources, but the departures from geodetic

gravity reduction become significant for regional surveys (Heck, 1990; Featherstone, 1995a).

Also of significance for the geophysical reduction of gravity data is the advent of positioning with the Global Positioning System or GPS (Featherstone, 1995b). It is essential to ensure that the correct horizontal and vertical coordinate systems are used during data reduction. For example, coordinate systems adopted for local surveying and mapping may not be geocentric, as is assumed in the reduction equations. If inappropriate coordinate systems are used, artifacts may be introduced into the data which can adversely affect the derived geological models (Featherstone, 1993). The correct use of coordinate systems is also essential for the correct integration of GPS-positioned gravity surveys with existing data.

THE GRAVITY ANOMALY

In geodesy, the gravity anomaly (Δg) is defined as the scalar difference between the Earth's gravity on the geoid (g_P) and normal gravity on the surface of the reference ellipsoid (γ) at the observation latitude (see Figure 1). The geoid is defined as the equipotential surface of the Earth's gravity field which corresponds most closely with mean sea-level. Thus,

$$\Delta g = g_P - \gamma \tag{1}$$

Geodesy requires gravity anomalies defined on the geoid for the solution of the boundary value problem of physical geodesy, which is used to determine the Figure of the Earth (Heiskanen and Moritz, 1967).

In geophysics, the gravity anomaly is used slightly differently. It is often defined as the scalar difference between the Earth's gravity on an arbitrarily defined, geoid-related vertical datum (g_Q) and normal gravity on the surface of the reference ellipsoid at the observation latitude. This is equivalent to using gravity data given on any other (non-geoid) equipotential surface of the Earth's gravity field (see Figure 1). Thus,

$$\Delta g = g_Q - \gamma \tag{2}$$

The geophysical requirements are not so stringent because only lateral variations in the anomalous gravity field are important. Moreover, it is preferable to use the mean elevation above the geoid as the vertical datum. This reduces the magnitude of the gravity reductions, thus making them less sensitive to errors from any assumptions

made, such as the bulk density of the local rocks. Equations (1) and (2) become identical at sea when the vertical datum for the gravity anomalies is inevitably chosen as the geoid.

GRAVITY REDUCTION

The reduction of surface gravity data (g_S) to g_P (on the geoid) or g_Q (another equipotential surface) removes the gravitational effects of topography and distance from the geocentre. The reduction itself involves the application of a series of corrections. These corrections account for the vertical gradient of gravity near the Earth's surface, the gravimetric attraction of the topography, and the centrifugal acceleration and oblate ellipticity of the figure of the Earth.

The following sections summarise the formulae used in geodesy and geophysics to reduce surface gravity observations to yield simple Bouguer gravity anomalies. These comprise the latitude correction or normal gravity (γ), the free-air correction (δg_F), and the simple or slab Bouguer correction (δg_B). The simple Bouguer anomaly is given by:

$$\Delta g_B = g_S - \gamma + \delta g_F - \delta g_B \quad (3)$$

NORMAL GRAVITY: THE LATITUDE CORRECTION

The latitude correction is intended to eliminate the centrifugal acceleration that affects observed gravity, and which is a function of latitude. It also accounts for the oblate elliptical shape of the Earth. The latitude correction is usually calculated from an international gravity formula (IGF), whose constants are based upon the mean Earth ellipsoid adopted by the International Association of Geodesy (IAG). This mean Earth ellipsoid is chosen such that its defining physical and geometrical parameters closely model those of the Earth (Chovitz, 1981). The most recent mean Earth ellipsoid adopted by the IAG is the Geodetic Reference System 1980 or GRS80 (Moritz, 1980), and which supersedes the Geodetic Reference System 1967 (GRS67).

In geodesy, a closed equation, called Somigliana's formula (Moritz, 1980), is used to determine normal gravity on the surface of the mean Earth ellipsoid:

$$\gamma = \gamma_a \frac{1 + k \sin^2 \phi}{\sqrt{1 - e^2 \sin^2 \phi}} \quad (4)$$

where k is the normal gravity constant, γ_a is normal gravity on the equator, e^2 is the square of the first numerical eccentricity, and ϕ is the geodetic latitude on the mean

Earth ellipsoid. The numerical values of these constants for GRS67 and GRS80 in Table 1 have been taken directly from International Association of Geodesy (1971) and Moritz (1980), respectively.

Table 1. The physical and geometrical constants required to compute normal gravity on the mean Earth ellipsoid when using Somigliana's closed formula.

	GRS67	GRS80
γ_a (ms ⁻²)	9.780 318 455 8	9.780 326 771 5
k	0.001 931 663 383	0.001 931 851 353
e^2	0.006 694 605 328 56	0.006 694 380 022 90

In geophysics, two less accurate Chebyshev approximations of equation (4) are utilised to estimate normal gravity. The first has a relative accuracy of 10^{-3} μms^{-2} (Moritz, *ibid.*), and is given by:

$$\gamma = \gamma_a(1 + \alpha \sin^2 \phi + \alpha_1 \sin^4 \phi + \alpha_2 \sin^6 \phi + \alpha_3 \sin^8 \phi) \quad (5)$$

The coefficients in equation (5) for GRS67 and GRS80 are listed in Table 2. Note that the value of normal gravity at the equator need only be given to nine decimal places in this approximation. The GRS67 coefficients were derived from the GRS67 constants e^2 and k in Table 1 using the equations given by Moritz (1980).

Table 2: The physical constants required to compute normal gravity using the eighth-order Chebyshev approximate formula.

	GRS67	GRS80
γ_a (ms ⁻²)	9.780 318 459	9.780 326 772
α	0.005 278 966 0	0.005 279 041 4
α_1	0.000 023 272 5	0.000 023 271 8
α_2	0.000 000 126 2	0.000 000 126 2
α_3	0.000 000 000 7	0.000 000 000 7

The second Chebyshev approximation (equation 6) has a relative accuracy of only 1 μms^{-2} (Moritz, *ibid.*), hence the need for only six decimal places in the definition of equatorial normal gravity in Table 3. This approximation is often referred to as the International Gravity Formula (IGF), and is the most common formula used to compute the latitude correction in geophysics; see for example, Telford *et al.* (1990). This is

despite the fact that the reading error of a gravimeter is typically one order of magnitude less than the approximation error associated with this formula.

$$\gamma = \gamma_a (\beta \sin^2 \phi + \beta_1 \sin^2 2\phi) \quad (6)$$

Table 3. The physical constants required to compute normal gravity using the second-order Chebyshev approximate formula.

	GRS67/IGF67	GRS80/IGF80
γ_a (ms ⁻²)	9.780 318	9.780 327
β	0.005 302 4	0.005 302 4
β_1	-0.000 005 9	-0.000 005 8

In some instances, the IGF30 (Cassinis, 1930) is still used to compute the latitude correction, whose defining constants are $\gamma_a = 9.780\,490\text{ ms}^{-2}$, $\beta = 0.005\,2884$, and $\beta_1 = 0.000\,0059$. As GRS80 is the more up-to-date and internationally adopted reference gravity field, it should be used in preference to a formula that was derived from surface gravity data collected over 65 years ago. Also, equation (4) is exact and is easily calculated on a computer with only a ~10% increase in computation time over equation (6). Therefore, equation (4) with GRS80 constants should be used in preference.

The atmospheric gravity correction

An additional consideration is that the parameters that define GRS67 and GRS80 were determined using predominantly satellite-derived geodetic data. As such, the computed normal gravity includes a component due to the mass of the Earth's atmosphere, whereas gravity observed on or close to the Earth's surface does not. The atmospheric gravity correction (δg_A) is added to the gravity anomalies so as to account for this.

$$\delta g_A = 8.71 - 1.03 \times 10^{-3} H \text{ ms}^{-2} \quad (7)$$

Equation (7) was derived from a least squares fit to the mean of the United States and Cospar International Reference Atmosphere models given in Ecker and Mittermayer (1969) and the International Association of Geodesy (1971). This correction term is typically small for geophysical surveys, and usually insignificant when compared to errors in the estimates of topographic density used for the Bouguer reduction. In effect, the atmospheric correction term can be considered to be a bias

that is unimportant to geophysical exploration, but is important in geodesy where the mass of the Earth is to be preserved.

The coordinate systems required for gravimetry

An ellipsoid is used as the reference surface for horizontal geodetic coordinates, namely latitude (ϕ) and longitude (λ). Of great importance is that the geodetic latitude of gravity observations used in equations (4), (5) and (6) be given with respect to a geocentric, mean Earth ellipsoid.

Historically, different ellipsoids have been chosen in different parts of the world in order to simplify surveying and mapping in that region, and these ellipsoids are not necessarily geocentric. As such, a single ground point can have more than one set of geographic coordinates by virtue of the ellipsoid used. For example, map-read geodetic coordinates in some countries can differ from geocentric coordinates by up to 1km. This is significant when compared to the acceptable error in the latitude (± 10 metres), which propagates an error in the latitude correction commensurate with the typical reading error of a gravimeter ($\pm 0.1 \mu\text{ms}^{-2}$).

Failure to use the geocentric coordinate system, as required by normal gravity, causes the propagation of systematic errors into the gravity anomalies (Featherstone, 1993). More importantly, co-registration errors occur when combining gravity data coordinated and reduced with respect to different ellipsoids. This effect becomes more pronounced for detailed gravity surveys, and is especially important now that the Global Positioning System (GPS) is commonly used as the primary means to coordinate gravity surveys (eg. Featherstone and Dentith, 1994).

COORDINATE TRANSFORMATIONS

GPS readily provides geocentric coordinates that are directly compatible with GRS80. Therefore, the GPS-derived geodetic latitude can be used immediately in equations (4), (5) or (6). However, where the coordinates are not geocentric, the transformation from a local ellipsoid to a geocentric ellipsoid can be achieved through the sequential use of three distinct calculations:

1. Convert the ellipsoidal coordinates (ϕ - geodetic latitude, λ - geodetic longitude, and h - ellipsoidal height) to Cartesian coordinates (X, Y, Z), centred on the local ellipsoid (A).
2. Transform these Cartesian coordinates to geocentric Cartesian coordinates using a seven-parameter conformal transformation.

3. Convert these Cartesian coordinates to geodetic coordinates (ϕ and λ only) based on the geocentric ellipsoid (B).

Ellipsoidal to Cartesian coordinate conversion

From Figure 2, three-dimensional geometry can be used to derive the conversion of ellipsoidal coordinates to Cartesian coordinates, which are centred on the *same* ellipsoid, and whose axes are aligned with the minor and major axes of that ellipsoid. These formulae, also given in Heiskanen and Moritz (1967), are:

$$X_A = (\nu + h) \cos \phi \cos \lambda \tag{8}$$

$$Y_A = (\nu + h) \cos \phi \sin \lambda \tag{9}$$

$$Z_A = (\nu(1 - e^2) + h) \sin \phi \tag{10}$$

where the radius of curvature of the ellipsoid in the prime vertical is:

$$\nu = \frac{a}{\sqrt{1 - e^2 \sin^2 \phi}} \tag{11}$$

and the constants a and e are those associated with the local ellipsoid. These data are usually available from local surveying and mapping authorities or may appear listed in Tables 7.4 and 7.5 of Defense Mapping Agency (1987).

Notice that the ellipsoidal height (h) is required in equations (8), (9) and (10). Unfortunately, this coordinate is not always readily available. However, using the height of the gravity observation above the geoid (H) instead, introduces an error of only a few centimetres, which is considerably less than the uncertainty of the seven transformation parameters (see below).

Seven-parameter transformation

The seven-parameter transformation is applied to the three-dimensional Cartesian coordinates resulting from equations (8), (9) and (10). It comprises an origin shift in three dimensions (ΔX , ΔY , ΔZ), a rotation about each coordinate axis (r_x , r_y , r_z), and a change in scale (ds), which is usually expressed in parts per million (ppm). These are applied to the Cartesian coordinates using matrix algebra:

$$\begin{pmatrix} X \\ Y \\ Z \end{pmatrix}_B = \begin{pmatrix} \Delta X \\ \Delta Y \\ \Delta Z \end{pmatrix} + (1+ds) \begin{pmatrix} 1 & r_z & -r_y \\ -r_z & 1 & r_x \\ r_y & -r_x & 1 \end{pmatrix} \begin{pmatrix} X \\ Y \\ Z \end{pmatrix}_A \quad (12)$$

Equation (12) is an approximate formula that holds for small axial rotations (typically less than five arc seconds), which usually apply when transforming coordinates between ellipsoids. The appropriate set of parameters is usually available from local surveying and mapping authorities. These can usually provide coordinate transformation accurate to less than five metres. However, the complete set of seven parameters is not always available and three origin shifts can be used in equation (12) instead. These may be listed in Table 7.5 of Defense Mapping Agency (1987) and can provide coordinate transformations accurate to approximately 20 metres. In this case, equation (12) reduces to:

$$\begin{pmatrix} X \\ Y \\ Z \end{pmatrix}_B = \begin{pmatrix} \Delta X \\ \Delta Y \\ \Delta Z \end{pmatrix} + \begin{pmatrix} X \\ Y \\ Z \end{pmatrix}_A \quad (13)$$

Cartesian to geodetic conversion

The conversion of Cartesian coordinates to geodetic coordinates can be achieved by inverting equations (8), (9) and (10). For the longitude, this gives:

$$\lambda = \tan^{-1} \frac{Y_B}{X_B} \quad (14)$$

However, to determine the geodetic latitude from a simple inversion requires several iterations and an improved equation was developed by Bowring (1976) and has been shown to converge more rapidly by Laskowski (1991). This is:

$$u = \tan^{-1} \left(\frac{Z_B b + \varepsilon^2 b^2 \sin^3 u}{a \sqrt{X_B^2 + Y_B^2 - e^2 a^2 \cos^3 u}} \right) \quad (15)$$

where u is the reduced latitude, which is related to the geodetic latitude by:

$$\tan \phi = \frac{a}{b} \tan u \quad (16)$$

and ε^2 is the square of the second numerical eccentricity and b is the length of the semi-minor axis of the ellipsoid.

Equation (15) is solved iteratively using

$$u = \tan^{-1} \left(\frac{Z_B a}{b \sqrt{X_B^2 + Y_B^2}} \right) \quad (17)$$

as a first approximation, then u is inserted in equation (16) to produce the geodetic latitude.

Equation (18) is included below for the sake of completeness, but must not be used for gravity reduction, which requires heights that are physically related to the geoid.

$$h = \sqrt{X_B^2 + Y_B^2} \cos \phi + Z_B \sin \phi - \frac{a^2}{\nu} \quad (18)$$

The geometrical constants that must be used in equations (15), (16), (17) and (18) are listed in Table 4.

Table 4. Geometrical constants required to compute geocentric geodetic coordinates.

	GRS67	GRS80
a (m)	6 378 160	6 378 137
b (m)	6 356 774.5161	6 356 752.3141
e^2	0.006 694 605 328 56	0.006 694 380 022 90
ε^2	0.006 739 725 128 32	0.006 739 496 775 48

The GPS height transformation

If GPS is used to coordinate a gravity survey, the height transformation is essential to avoid coordinate-system-related gravity artifacts (Featherstone, 1993). GPS provides purely geometrical heights above the surface of the GRS80 ellipsoid, which have no physical meaning as they are not referred to the equipotential surfaces of the Earth's gravity field. Therefore, they must not be used for gravity reductions. Firstly, the GPS ellipsoidal heights must be transformed to geoid-related heights, as would have been derived from conventional geodetic levelling.

As the geoid undulates with respect to the ellipsoid, ironically due to the effects of gravity, the geoid-ellipsoid separation (N), and hence the transformation, depends upon

geographical location. From Figure 1, the relationship between GPS ellipsoidal height (h) and geoid-related height (H) is given algebraically by:

$$H = h - N \quad (19)$$

Geoid-ellipsoid separations are also usually available from local surveying and mapping authorities. However, the position of the geoid is not always precisely known and may be in error by a few metres.

THE FREE-AIR GRAVITY REDUCTION

The free-air reduction accounts for gravity observations not made on the vertical datum surface. This is accounted for using the vertical gravity gradient as if the observation were made in free-air a distance H above or below the vertical datum surface. It is essentially a correction to the observed gravity for the inverse distance-squared decay of gravity on moving away from the Earth. The linear approximation, based upon a spherical Earth model, often employed in geophysics, is:

$$\delta g_F = \frac{2\bar{g}}{R} H \quad (20)$$

where \bar{g} is mean gravity on an assumed spherical Earth of radius R . The commonly adopted numerical value of the free-air gradient is $3.086 \mu\text{ms}^{-2}\text{m}^{-1}$ (eg. Telford *et al.*, 1990). The free-air reduction is added to observed gravity for observations above the vertical datum surface and subtracted for those below.

In geodesy, this spherical Earth approximation is considered inadequate because the figure of the Earth is more accurately represented by an oblate ellipsoid. Therefore, a second-order correction, based upon a Taylor expansion of normal gravity above the Earth, is used. This second-order free-air reduction is derived in Featherstone (1995a) as:

$$\delta g_F = \frac{2\gamma}{a}(1 + f + m - 2f \sin^2 \phi)H - \frac{3\gamma}{a^2}H^2 \quad (21)$$

where f is the geometrical flattening of the mean Earth ellipsoid, m is the geodetic parameter, which is the ratio of gravitational and centrifugal forces at the equator, and all other quantities are as defined earlier; also see Table 5. The numerical value of normal gravity in equation (21) should be computed using equation (4).

Table 5. Physical and geometrical constants required to compute the second-order free-air reduction.

	GRS67	GRS80
a (m)	637 816 0	637 813 7
f	0.003 352 923 712 99	0.003 352 810 681 18
m	0.003 449 801 434 30	0.003 449 786 003 08

For instance, the difference between the linear and second-order free-air reductions (equation 21 minus equation 20) reaches a maximum of $-4.986 \mu\text{ms}^{-2}$ at the summit of Mount Everest ($\phi \approx 27^\circ 58'$ and $H \approx 8848$ metres).

THE BOUGUER GRAVITY REDUCTION

The free-air reduction neglects the attraction of the topography between the Earth's surface and vertical datum surface. Essentially, the free-air reduction is effectively a condensation reduction, where all topographic masses are condensed onto the chosen equipotential surface. The Bouguer reduction accounts for the intermediate topography in two stages. Firstly, the simple Bouguer reduction (δg_B) assumes that the topography attracts as an infinitely lateral plate of thickness equal to the observation elevation, and is given by:

$$\delta g_B = 2\pi G\rho H \quad (22)$$

where G is the Newtonian gravitational constant, and ρ is the mean topographic bulk density, which is often assumed to be 2670 kgm^{-3} , an approximate crustal average.

The second stage accounts for departures of the topography from this simple plate approximation, and takes into account the negative gravitational effect of the residual topography. Thus, the gravimetric terrain correction is always positive and is added to the simple Bouguer anomaly in equation (3). This terrain correction term is only pertinent in topographically rugged areas and is typically one tenth that of the plate reduction. The computation of the terrain correction has been discussed previously by LaFehr (1991) and Ma and Watts (1994) and is thus not considered here.

A GENERALISED REDUCTION FORMULA

Equations (3), (4), (21) and (22) have been combined to give the following generalised formula, which applies to both geodesy and geophysics depending upon the choice of vertical datum. In geodesy, H refers to the geoid, and usually in geophysics, H refers to

the arbitrarily defined equipotential surface. Normal gravity should always be computed via Somigliana's formula (equation 4), and must use the geocentric latitude as its argument. The numerical values of the GRS80 geometrical and physical constants should be taken from Tables 1 and 5. The combination yields:

$$\Delta g_B = g_S - \gamma \left(1 - \frac{2H}{a} \left\{ 1 + f + m - \frac{\pi G \rho a}{\gamma} - 2f \sin^2 \phi \right\} + \delta \frac{3H^2}{a^2} \right) \quad (23)$$

where $\delta = +1$ for $H > 0$ and $\delta = -1$ for $H < 0$. For $H = 0$ equations (1) and (2) apply, which are identical at the geoid. Thus,

$$\Delta g = g_P - \gamma = g_Q - \gamma = g_S - \gamma \quad (24)$$

THE FORTRAN PROGRAMS

Two subroutines, written in FORTRAN77, are available for public access on the *Computers & Geosciences* server (www.iamg.org). The subroutine COORTRAN transforms the geodetic coordinates of existing gravity data to a geocentric ellipsoid, given the appropriate seven transformation parameters and local ellipsoidal constants. The second subroutine, GRAVRED, applies the generalised reduction formulae (23) and (24), together with the Somigliana closed formula (4) for GRS80 normal gravity.

These subroutines were originally developed to transform and reduce gravity data from the Australian Geodetic Datum to GRS80 (Featherstone, 1995a), and thus uses the ellipsoidal constants and transformation parameters for Australia. To apply these subroutines elsewhere requires the relevant ellipsoidal constants and transformation parameters, which are usually available from local surveying and mapping authorities or Tables 7.4 and 7.5 of Defense Mapping Agency (1987).

An obvious test for the COORTRAN subroutine is to transform the same set of coordinates between two coordinate systems repeatedly. The reverse transformation is achieved by changing the signs of the transformation parameters used in equations (12) and (13). The computer rounding errors in FORTRAN77 double precision mode are less than a millimetre per complete transformation and do not propagate quickly during repeat forward and reverse transformations. These errors are numerically divergent, but as the coordinate transformation is usually only required once, millimetre-level numerical accuracy can be routinely achieved. However, the accuracy of the transformed coordinates ultimately depends on the accuracy of the transformation parameters used.

The Bowring (1976) algorithm converges to 1 part in 10^{10} after only three iterations, whereas iteration using the inverse of equations (8), (9) and (10) converges after ten iterations. This can result in a three-fold reduction in computation time when transforming coordinates for a large gravity data-base.

Implementation of the programs

The subroutines can be easily applied either to newly collected gravity data or to transform existing gravity data-bases. Many existing gravity data-bases are referred to GRS67, or in some cases IGF30. Ideally, these should be updated to the modern and internationally accepted GRS80, preferably using the approaches described herein. However, to update existing Bouguer anomalies to GRS80 and apply the second-order free-air reduction requires the original surface gravity observations, elevations and geodetic coordinates.

If GPS is used to coordinate a gravity survey, the geodetic latitude provided by GPS can be used directly in equation (4) without the need for transformations. However, the GPS-derived ellipsoidal height must be converted to a geoid-related height by applying equation (18). This is essential because gravity reduction is highly susceptible to elevation errors. For example, a ± 0.02 metre elevation uncertainty propagates as $\pm 0.1 \mu\text{ms}^{-2}$ in the combined free-air and simple Bouguer reductions, assuming an accurate estimate of the topographic density.

A CASE STUDY IN WESTERN AUSTRALIA

An 80.1 kilometre geodetic gravity profile in south-west Western Australia has been chosen to test the geodetic gravity reductions, as coordinates determined using both GPS and conventional survey methods were readily available (Featherstone, 1993). This provides a sound geodetic framework in which to analyse the effect of different coordinate systems on the reduction of gravity data. The profile is also considered typical of a regional scale gravity survey for the identification of sub-surface structure. However, this applies equally to all scales and extents of gravity survey.

Figure 3 illustrates the numerical differences among simple Bouguer anomalies reduced from the same observational data using the following three sets of coordinates:

1. GPS-derived WGS84 ellipsoidal coordinates alone, denoted by $(\phi_{\text{WGS84}}, h_{\text{WGS84}})$;
2. Australian coordinates on the Australian Geodetic Datum 1984 (AGD84) and the Australian Height Datum (AHD), denoted by $(\phi_{\text{AGD84}}, h_{\text{AHD}})$;
3. WGS84 geographical coordinates and the geoid-derived AHD elevation, determined using equation (19), denoted by $(\phi_{\text{WGS84}}, h_{\text{WGS84}} - N_{\text{WGS84}})$.

In Figure 3, profile (c) shows the geodetically reduced gravity data, using equation (23) with GPS-derived WGS84 ellipsoidal coordinates in conjunction with the AUSGEOID93 Australian geoid model (Steed and Holtznagel, 1994). At the scale of Figure 1, there is little resolvable difference between profiles (b) and (c). Nevertheless, a small difference does exist which varies approximately linearly from $0.9 \mu\text{ms}^{-2}$ and $1.1 \mu\text{ms}^{-2}$, and is predominantly due to the difference in normal gravity due to the use of AGD84 geodetic latitude in equation (4). This difference can increase five-fold for other geodetic coordinate systems, and often larger than the typical reading error of a gravimeter.

For the purposes of a localised geophysical survey, this can be treated as a bias or regional trend, as it is only the residual Bouguer gravity anomaly which gives detailed information of near-surface geological structure. Therefore, the choice of geodetic latitude is not necessarily critical in the data reduction for a single survey. However, it must be considered if GPS-derived and existing data are to be integrated with one another.

Of more interest is profile (a), in which the GPS-derived WGS84 ellipsoidal coordinates alone are employed to reduce the gravity data. The most significant difference exists in the shape of the gravity anomaly, which is illustrated more clearly in Figure 4. The difference is irregular and uncorrelated with the Bouguer anomalies in Figure 3. It varies between $54.1 \mu\text{ms}^{-2}$ and $55.5 \mu\text{ms}^{-2}$ and causes errors in both the high and low frequency components of the gravity anomaly.

The differences displayed in Figure 4 are due to the susceptibility of the free-air and Bouguer reductions to changes in the vertical datum surface. The profile in Figure 4 is equivalent to the application of the free-air and Bouguer reductions to the geoid-ellipsoid separation (N). It is, therefore, evident that the deduced geological model will be different for profiles (a) and (c) in Figure 1, especially when attempting to delineate the more subtle geological features. For this reason alone, the rigorous equation (23) should be employed for gravity data reduction, and especially so when GPS is used to provide vertical control for a gravity survey.

For the 30 gravity observations used in this example, the exact geodetic reduction only adds an extra ~10% to the computation time on a Sun Sparcstation model 30, when compared to the conventional geophysical gravity reduction. This illustrates that the precise geodetic reduction of gravity data collected for geophysical purposes can now be applied without the need for the approximations used only for historical reasons.

CONCLUDING REMARKS

Gravity data collected for geophysical purposes should be reduced using the precise geodetic approach described herein. This is easily implemented, does not introduce a significant increase in computation time, and avoids the approximations used in the geophysical reduction formulae.

Of most importance is the correct use of coordinate systems for gravity reduction. This is necessary now that the Global Positioning System is widely used to provide three-dimensional coordination of gravity surveys. Geocentric geodetic coordinates are required to apply meaningful gravity reductions and to avoid the introduction of artifacts into the gravity anomalies, which will alter the geological models derived from these data.

Moreover, coordinate transformations are required for the correct co-registration of GPS positioned gravity data with existing data.

ACKNOWLEDGEMENTS

We would like to thank *Adare Geophysics* and *Geodetic Solutions* for their permission to publish this paper. Thanks are also extended to the two anonymous reviewers for their time taken to consider this manuscript.

REFERENCES

- Bowring, B.R., 1976, Transformation from spatial to geographical coordinates, *Survey Review*, v. 23, no. 181, p. 323-327.
- Cassinis, G., 1930, Sur l'adoption d'une formule internationale pour la pesanteur normale, *Bulletin Géodésique*, v. 26, 40-49.
- Chovitz, B.H., 1981, Modern geodetic Earth reference models, *EOS - Transactions of the American Geophysical Union*, v. 62, no. 7, p. 65-67.
- Defense Mapping Agency, 1987, Department of Defense World Geodetic System 1984: its definition and relationships with local geodetic systems. Technical Report no. 8350.2, Washington, 121p.
- Ecker, E. and Mittermayer, E., 1969, Gravity corrections for the influence of the atmosphere, *Bollettino di Geofisica Teoretica et Applicata*, v. 21, no. 41, p. 70-80.
- Featherstone, W. E., 1993, GPS coordinate transformations and their use in gravimetry, *Exploration Geophysics*, v. 24, no. 4, p. 487-492.
- Featherstone, W.E. and Dentith, M.C., 1994, Matters of gravity: the search for gold, *GPS World*, v.5, no. 7, p. 34-39.
- Featherstone, W.E., 1995a, On the use of Australian geodetic datums in gravity field determination, *Geomatics Research Australasia*, no. 62, p. 19-29.

- Featherstone, W.E., 1995b, The Global Positioning System and its use in geophysical exploration, *Exploration Geophysics*, v. 26, no. 1, p. 1-18.
- Heck, B., 1990, An evaluation of some systematic error sources affecting terrestrial gravity anomalies, *Bulletin Géodésique*, v. 64, no. 2, p. 88-108.
- Heiskanen, W.A. and Moritz, H. (1967) *Physical Geodesy*, Freeman, San Francisco, 364p.
- International Association of Geodesy, 1971, Geodetic Reference System 1967, Special Publication no. 3 du *Bulletin Géodésique*, 116p.
- LaFehr, T. R., 1991, Standardization in gravity reduction, *Geophysics*, v. 56, no. 8, p. 1170-1178.
- Laskowski, P., 1991, Is Newton's iteration faster than simple iteration for transformation between geocentric and geodetic coordinates?, *Bulletin Géodésique*, v. 65, no. 1, p. 14-17.
- Ma and Watts, 1994, Terrain correction program for regional gravity surveys. *Computers and Geosciences*, v. 20, no. 10, p. 961-972.
- Moritz, H., 1980, Geodetic Reference System 1980, *Bulletin Géodésique*, v. 54, p. 395-405.
- Steed, J. and Holtznagel, S. (1994) AHD heights from GPS using AUSGEOID93, *The Australian Surveyor*, v. 39, no. 1, p. 21-27.
- Telford, W.M., Geldart, L.P. and Sheriff, R.E., 1990, *Applied Geophysics* (second edition), Cambridge University Press, Cambridge, 770p.

APPENDIX 1

program coortran

```
SUBROUTINE COORTRAN(ALAT,ALON,HT,WLAT,WLON)
  IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C
C A FORTRAN77 PROGRAM FOR THE CONVERSION OF GEOGRAPHICAL COORDINATES
C FROM A LOCAL ELLIPSOID TO GRS80, USING EITHER A SEVEN-PARAMETER OR
C THREE-PARAMETER TRANSFORMATION.
C
C INPUT:- LATITUDE (ALAT),LONGITUDE (ALON) AND HEIGHT (HT) ON LOCAL
C ELLIPSOID. HT MAY BE ABOVE GEOID IF ELLIPSOIDAL HEIGHT UNAVAILABLE
C OUTPUT: LATITUDE (TLAT) AND LONGITUDE (TLON) ON GRS80
C (AL IN RADIANS)
C
  DRCON=3.1415926535897932384626433832D0/180.0D0
C
C GRS80 GEOMETRICAL CONSTANTS (MORITZ, 1980)
C
  WA=6378137.D0          ! SEMI MAJOR AXIS LENGTH (METRES)
  WB=6356752.3141D0     ! SEMI MINOR AXIS LENGTH (METRES)
  WE2=6.69438002290D-3  ! 1ST ECCENTRICITY SQUARED
  W2E2=6.73949677548D-3 ! 2ND ECCENTRICITY SQUARED
C
C SEVEN TRANSFORMATION PARAMETERS FROM AUSTRALIAN GEODETIC DATUM
C TO GRS80. IF THREE PARAMETERS ARE ONLY AVAILABLE, SET ARX ARY
C ARZ AND ADS TO ZERO.
C
  AXO=-116.00D0
  AYO=-50.47D0          ! 3D ORIGIN SHIFTS (METRES)
  AZO=+141.69D0
  ARX=(-0.230D0/3600.D0)*DRCON
  ARY=(-0.390D0/3600.D0)*DRCON ! ROTATIONS (RADIANS)
  ARZ=(-0.344D0/3600.D0)*DRCON
  ADS=1.0D0+0.09830D-6 ! SCALE CHANGE
C
C AUSTRALIAN ELLIPSOIDAL CONSTANTS
C
  AA=6378160.D0          ! SEMI MAJOR AXIS LENGTH (METRES)
  AE2=6.6945418545D-3   ! 1ST ECCENTRICITY SQUARED
C
  CALL FWDBWG(ALAT,ALON,HT,AA,AE2,AX,AY,AZ)
  CALL SEVPRM(AX,AY,AZ,AXO,AYO,AZO,ARX,ARY,ARZ,ADS,WX,WY,WZ)
  CALL REVBWG(WX,WY,WZ,WA,WB,W2E2,WLAT,WLON)
  RETURN
  END
C
C *****
C
  SUBROUTINE FWDBWG(OLAT,OLON,HL,A,E2,X,Y,Z)
  IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C
C CONVERSION OF GEOGRAPHICAL TO CARTESIAN COORDINATES ON INITIAL
C ELLIPSOID (HEISKANEN AND MORITZ, 1967)
C
  V=A/(DSQRT(1.D0-(E2*(DSIN(OLAT))**2)))
  X=(V+HL)*DCOS(OLAT)*DCOS(OLON)
  Y=(V+HL)*DCOS(OLAT)*DSIN(OLON)
  Z=(V*(1.D0-E2)+HL)*DSIN(OLAT)
  RETURN
  END
C
C *****
C
  SUBROUTINE SEVPRM(X,Y,Z,XO,YO,ZO,RX,RY,RZ,DS,X1,Y1,Z1)
  IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C
C SEVEN-PARAMETER TRANSFORMATION BETWEEN 3-D CARTESIAN COORDINATES
```

```

C   FOR SMALL AXIAL ROTATIONS (LESS THAN FIVE ARC SECONDS)
C
      X1=(DS*(X-(RY*Z)+(RZ*Y)))+XO
      Y1=(DS*(Y-(RZ*X)+(RX*Z)))+YO
      Z1=(DS*(Z-(RX*Y)+(RY*X)))+ZO
      RETURN
      END
C
C *****
C
      SUBROUTINE REVBWG(X,Y,Z,A,B,E2,E22,OLAT,OLON)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C
C   ITERATIVE CONVERSION OF CARTESIAN TO GEOGRAPHICAL COORDINATES
C   ON THE GRS80 ELLIPSOID (BOWRING, 1976)
C
      PI=3.1415926535897932384626433832D0
      OLON=DATAN(Y/X)
      IF (OLON.LT.0.D0) OLON=OLON+PI
C
      PEE=DSQRT(X**2+Y**2)
      FRSTU=DATAN((A/B)*(Z/PEE))
30   AU=DATAN((B*(Z+(B*E22*(DSIN(FRSTU))**3)))/
      .   (A*(PEE-(A*E2*(DCOS(FRSTU))**3))))
      ALTN=ABS(AU-FRSTU)
      IF (ALTN.GT.1.D-10) THEN
      FRSTU=AU
      GOTO 30
      END IF
      OLAT=DATAN(A*DTAN(AU)/B)
      RETURN
      END
C
C *****

```

program gravred

```

SUBROUTINE GRAVRED(WLAT,HT,GRAV,RHO,BANOM)
  IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C
C   PRODUCES SIMPLE BOUGUER ANOMALIES WRT GRS80 FROM SURFACE GRAVITY
C   OBSERVATIONS USING THE SOMIGLIANA CLOSED FORMULA FOR NORMAL
C   GRAVITY AND A SECOND-ORDER FREE-AIR REDUCTION
C
C   INPUT:- GRS80 LATITUDE IN RADIANS(WLAT), GEOID-RELATED HEIGHT IN
C   METRES (HT), OBSERVED GRAVITY IN MGAL (GRAV), MEAN TOPOGRAPHIC
C   DENSITY IN KG/M**3 (RHO).
C   OUTPUT:- SIMPLE BOUGUER ANOMALY IN MGAL (BANOM)
C
      DATA PI,BIGG/3.1415926535897932384626433832D0,6.67259D-11/
C
C   GRS80 GEOMETRICAL AND PHYSICAL CONSTANTS (MORITZ, 1980)
C
      WA=6378137.D0      ! SEMI MAJOR AXIS LENGTH (METRES)
      WF=3.35281068118D-3 ! GEOMETRICAL FLATTENING
      WM=3.44978600308D-3 ! GEODETIC PARAMETER
      WK=1.931851353D-3   ! NORMAL GRAVITY CONSTANT
      WE2=6.69438002290D-3 ! 1ST ECCENTRICITY SQUARED
      GAMA=978032.67715D0 ! NORMAL GRAVITY ON EQUATOR (MGAL)
C
      SLAT=DSIN(WLAT)
      S2LAT=SLAT**2
C
C   EVALUATE SOMIGLIANA CLOSED NORMAL GRAVITY (MGAL)
C
      DNORMG=GAMA*((1.D0+WK*S2LAT)/(DSQRT(1.D0-WE2*S2LAT)))
C
      DFHT=HT/WA
      BOU=(PI*BIGG*RHO*HT*WA)/DNORMG
      TSQR=2.D0*DFHT*(1.D0+WF+WM-BOU-2.D0*WF*S2LAT)
      TCUB=3.D0*(DFHT**2)
C
      IF (HT.GT.0.D0) THEN
        RED=1.D0-TSQR+TCUB
      ELSE IF (HT.LT.0.D0) THEN
        RED=1.D0-TCUB-TSQR
      ELSE IF (HT.EQ.0.D0) THEN
        RED=1.D0
      END IF
      CALL SRATM(CORATM,HT)
      BANOM=GRAV-DNORMG*RED+CORATM
      RETURN
      END
C
C *****
C
      SUBROUTINE SRATM(CORATM,HT)
        IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C
C   COMPUTES THE ATMOSPHERIC GRAVITY CORRECTION (MGAL)
C
      DATA ATA,ATB/0.871D0,1.03D-4/
      IF (HT .GT. 0.) THEN
        HT2=HT**2
        CORATM=ATA-ATB*HT
      ELSE IF (HT .LE. 0.) THEN
        CORATM=ATA
      END IF
      RETURN
      END
C
C *****

```

Figure 1. Surface gravity (g_S), geoidal gravity (g_P), normal gravity (γ) and gravity on an arbitrary equipotential surface (g_Q). N is the geoid-ellipsoid separation, H is the height relative to the geoid or any other equipotential surface, and h is the ellipsoidal height.

Figure 2. The geometrical relationship between Cartesian coordinates (X, Y, Z) and ellipsoidal coordinates (ϕ, λ, h).

Figure 3. Simple Bouguer anomalies derived from (a) WGS84 ellipsoidal coordinates alone (dashed line), (b) AGD84 geodetic coordinates and AHD heights (dotted line), and (c) WGS84 geodetic coordinates and GPS-geoid derived AHD heights (solid line).
(units in mgal)

Figure 4. Difference between simple Bouguer anomalies derived from WGS84 ellipsoidal coordinates alone and WGS84 geodetic coordinates with GPS-geoid derived AHD heights (units in mgal)

Table 1. The physical and geometrical constants required to compute normal gravity on the mean Earth ellipsoid when using Somigliana's closed formula.

	GRS67	GRS80
γ_a (ms ⁻²)	9.780 318 455 8	9.780 326 771 5
k	0.001 931 663 383	0.001 931 851 353
e^2	0.006 694 605 328 56	0.006 694 380 022 90

Table 2: The physical constants required to compute normal gravity using the eighth-order Chebyshev approximate formula.

	GRS67	GRS80
γ_a (ms ⁻²)	9.780 318 459	9.780 326 772
α	0.005 278 966 0	0.005 279 041 4
α_1	0.000 023 272 5	0.000 023 271 8
α_2	0.000 000 126 2	0.000 000 126 2
α_3	0.000 000 000 7	0.000 000 000 7

Table 3. The physical constants required to compute normal gravity using the second-order Chebyshev approximate formula.

	GRS67/IGF67	GRS80/IGF80
γ_a (ms ⁻²)	9.780 318	9.780 327
β	0.005 302 4	0.005 302 4
β_1	-0.000 005 9	-0.000 005 8

Table 4. Geometrical constants required to compute geocentric geodetic coordinates.

	GRS67	GRS80
a (m)	6 378 160	6 378 137
b (m)	6 356 774.5161	6 356 752.3141
e^2	0.006 694 605 328 56	0.006 694 380 022 90
ε^2	0.006 739 725 128 32	0.006 739 496 775 48

Table 5. Physical and geometrical constants required to compute the second-order free-air reduction.

	GRS67	GRS80
a (m)	637 816 0	637 813 7
f	0.003 352 923 712 99	0.003 352 810 681 18
m	0.003 449 801 434 30	0.003 449 786 003 08